
はじめに

 組込み向けLinux開発における要求・課題
 クロス開発環境の準備

 プログラム（パッケージ）間の問題解決

 依存、競合、バージョンごとの互換性

 最小構成の実現

 必須機能以外を省いた環境構築

 多重開発の防止

 共通化可能なリソースの発見と流用

 組込み向けLinuxビルドツール「Poky」を活用
 Pokyの概要説明

 自分向けにカスタマイズする

 使用上の課題を考える 対象バージョン：
Poky 6.0 edison (Yocto Project 1.1)



3

目次

 Pokyの概要
 全体像

 構成要素

 ビルドフロー

 ディレクトリ構造

 レシピの作成
 記述方法

 例）helloレシピ

 Pokyをカスタマイズする
 バージョンアップ

 カーネルレシピの作成

 BSPの作成

 使用上の課題



4

目次

 Pokyの概要
 全体像

 構成要素

 ビルドフロー

 ディレクトリ構造

 レシピの作成
 記述方法

 例）helloレシピ

 Pokyをカスタマイズする
 バージョンアップ

 カーネルレシピの作成

 BSPの作成

 使用上の課題



5

Pokyの概要

 組込み向けカスタムLinuxを構築可能なビルドシステム
 出力：カーネル、ユーザランド（rootfs）、ツールチェイン

 パッケージ単位のカスタマイズが容易

 Yocto Projectの主要コンポーネントの一つ

 Yocto Project (http://www.yoctoproject.org/)
 幅広いハードウェア向けにカスタムLinuxディストリビューションを
構築するための開発基盤・ツールを提供

 開発者の負荷を軽減、開発の複雑性を緩和

 対応アーキテクチャ
 x86(32bit & 64bit), ARM, PowerPC, MIPS

 GUIによる操作
 ビルド対象の選択・カスタマイズやビルドシーケンスの確認

 Eclipse経由でアプリケーション開発、デバッグが可能

http://www.yoctoproject.org/

全体像

Upstreamソースコード

Poky

共通ビルド定義

ボードA向け
ビルド定義

ボードB向け
ビルド定義

ボードC向け
ビルド定義

kernel A

rootfs A

toolchain A

kernel B

rootfs B

toolchain B

kernel C

rootfs C

toolchain C

Yocto Project ソースリポジトリ

fetch fetch

build build build

ビルドの手順

nativeツールをビルド

ターゲット向けバイナリをビルド
(アプリ、ライブラリ、カーネル)

クロスビルドツールをビルド

rootfsを構築

インタプリタ、SCM、アーカイバ、
パッケージングツール

gcc, binutils, libc

出来上がったバイナリは
一度パッケージングされる
(.deb, .rpm, .ipk)

必要なパッケージのみを
インストールして構築

ターゲット向けバイナリの生成に必要なツールは、
基本的にnative環境に依存しない

主な構成要素

 レシピファイル (*.bb)
 ビルド方法の定義書（テキスト）

 シェルスクリプト＋Pythonの混合記述

 ソースfetch～パッケージングまでの動作を定義

 レシピ同士の依存関係を定義

 bitbake

 ビルドツール（pythonスクリプト）

 レシピファイルを解析し、ビルドアクションを実行
 実行例)  $ bitbake busybox

 設定ファイル (*.conf)
 ターゲットマシン別のビルド定義

 アーキテクチャ指定（march等）、出力形式（uImage等）

 ユーザ独自設定

 ターゲットマシンの選択、並行ビルドの設定等

http://link.swc.toshiba.co.jp/wiki/index.php?title=File:Crystal_Clear_apps_tux.png&filetimestamp=20101005104009


9

ビルドフロー

*.conf

eglibc.bb

busybox.bb

openssh.bb

openssl.bb

gcc-cross.bb

binutils-cross.bb

rootfs.bb

xxx.bbxxx.bbhoge.bb

依存
target

cross
依存

native依存

*.conf*.conf
設定のロード

xxx.bbxxx.bbhoge

eglibc openssh

rootfs

ビルド

busybox openssl

gcc-cross

binutils-cross

bitbake

レシピ バイナリ

ビルド

ビルド

ビルド依存

ビルド依存

インストール

BSPによるターゲット依存部の分離

 BSP

 ターゲット依存の設定を含むレシピセット

 必要なBSPを再利用・組み合わせて独自環境をビルドできる

 ターゲット依存設定の例
 ターゲットボード（H/W）依存の設定

 カーネルソース、.config、イメージ形式（zImage, uImage, etc.）

 ブートローダのデフォルトパラメータ（ロードアドレス等）

 静的デバイスファイル、コンソール名（ttyAMA, ttyO, etc.）

 環境ごとにカスタマイズする設定

 /etc以下の設定

 rc, inittab, fstab

 各種デーモン設定

 configureオプション（.config）の変更

BSPによるターゲット依存部の分離

 全ボード共通レシピ＋ターゲット依存レシピ（BSP）の構成

 レイヤ構造

 各BSPが共通レシピに対し追加設定をオーバライド

Poky

共通レシピ ボード依存レシピ カスタマイズレシピ

ボード依存レシピA

ボードA

カスタマイズレシピX

ボード依存レシピB

ボードB

カスタマイズレシピY

ボード依存レシピB

カスタマイズレシピX

カスタマイズレシピZ

共通レシピ

設定上書き

パターンA-1 パターンB-1 パターンB-2

Pokyのディレクトリ構造

bitbake/

meta/

classes/

recipes-kernel/

recipes-core/

linux/

linux-libc-headers

*.bb*.bb

*.bbclass*.bbclass

*.bb*.bb

busybox/ *.bb*.bb

bitbakeバイナリ・ライブラリ

共有関数群

レシピ実体・付属ファイル

conf/ *.conf

各種設定ファイル

meta-bsp1/ conf/

recipes-core/

meta-bsp2/

*.conf

ボード依存の設定ファイル

busybox/

bsp-app/ *.bb*.bb

*.bbappend*.bbappend

設定追加・変更

ビルドディレクトリの構造とビルドフロー

$TARGET/

build-$MACHINE/

tmp/downloads/

work/ deploy/

$NATIVE/ $MACHINE/

rootfs/kernel/rootfs/kernel/yyy/
rootfs/kernel/xxx/

sysroots/

ビルドディレクトリの構造とビルドフロー

$TARGET/

build-$MACHINE/

tmp/downloads/

work/ deploy/

$NATIVE/ $MACHINE/

rootfs/kernel/rootfs/kernel/yyy/
rootfs/kernel/xxx/

sysroots/

fetch

ソースアーカイブ、リポジトリ、etc.

unpack

& build

ビルドディレクトリの構造とビルドフロー

$TARGET/

build-$MACHINE/

tmp/downloads/

work/ deploy/

$NATIVE/ $MACHINE/

rootfs/kernel/rootfs/kernel/yyy/
rootfs/kernel/xxx/

sysroots/

ビルドディレクトリ構造とビルドフロー

$TARGET/

build-$MACHINE/

tmp/downloads/

work/ deploy/

$NATIVE/ $MACHINE/

rootfs/kernel/rootfs/kernel/yyy/
rootfs/kernel/xxx/

sysroots/

ネイティブ用ツール、スクリプト、ターゲット用ライブラリ

ビルドディレクトリ構造とビルドフロー

$TARGET/

build-$MACHINE/

tmp/downloads/

work/ deploy/

$NATIVE/ $MACHINE/

rootfs/kernel/rootfs/kernel/yyy/
rootfs/kernel/xxx/

sysroots/

sysroots以下のツール・ライブラリを使用して
ターゲット用バイナリ・イメージをビルド

レシピの作成

 ライセンス定義

 タスク定義

 ソースファイル定義

 ビルド依存定義

 パッケージングの設定

 例: hello.bb

レシピの作成－ ライセンス定義

 LICENSE：ライセンス名

 LIC_FILES_CHKSUM: ファイル名＋md5

 COPYING 、LICENSE等

 .cや.hの一部分を同時に参照する場合もあり

 フォーマット：
file://FILENAME;beginline=n1;endline=n2;md5=CHECKSUM

LICENSE = "GPLv2"

LIC_FILES_CHKSUM = "file://COPYING;md5=1a2b3c..."

レシピの作成－ タスク定義

 bitbakeはタスク定義に従ってビルドコマンドを実行

 基本タスク

1. do_fetch: ソースのダウンロード（アーカイブ、gitリポジトリ、等）

2. do_unpack: ソースを展開

3. do_patch: ローカルパッチの適用

4. do_configure: configure実行

5. do_compile: make実行

6. do_install: make install実行

7. ...

 規定タスクは定義済み、必要に応じて上書き

 タスク間には依存関係あり

 独自タスクを追加定義可能

レシピの作成－ タスク定義（例）

do_patch() {

if [ "${PV}" -ge "3" ]; then

patch -p1 < ${WORKDIR}/extra.patch

fi

}

do_configure() {

:

}

do_install_append() {

install -d ${D}/${bindir}

install -m 0755 ${S}/extrafile ${D}/${bindir}

}

addtask hoge before do_configure after do_patch

do_hoge() {

......

}

prepend/appendで規定関数前/後に
処理を追加可能

空処理で上書き（configureが不要な場合等）

独自タスクを追加

別処理で上書き（条件を追加したい場合等）

レシピの作成－ ソースファイル定義

 SRC_URI：ソースファイルの一覧

 プロトコル：file, http, git, svn, cvs, etc.

 規定タスクの一部はSRC_URIを見て自動で処理を実行

 do_fetch: プロトコルを見てwget, gitなどのfetchコマンドを自動実行

 do_unpack: 拡張子を見て展開コマンドを自動実行（tar, git checkout）

 do_patch: *.patch、*.diffなどを自動適用

SRC_URI = " ¥

file://localfile.patch ¥

http://www.server.com/archive.tar.gz ¥

git://git.server.com/repo.git;protocol=git;tag=xxx ¥

"

レシピの作成－ ビルド依存定義

 DEPENDS: 事前にビルドしておく必要のあるレシピ一覧

 ビルドで必要なライブラリなどを事前に生成

 .debのBuild-dependsフィールドに該当

DEPENDS = "foo bar baz"

レシピの作成－パッケージングの設定

 ビルド済みファイルはPoky内部でパッケージングされる

 パッケージ形式：.deb、.rpm、.ipk

 PACKAGES：生成するパッケージ名一覧

 FILES_xxx：パッケージに含めるファイル一覧

 RDEPENDS_xxx: 依存パッケージ一覧

 .debのDependsフィールドに該当

PACKAGES = "hoge-bin hoge-lib hoge-conf"

FILES_hoge-bin = "${bindir}/*"

FILES_hoge-lib = "${libdir}/*"

FILES_hoge-conf = "${sysconfdir}/*"

RDEPENDS_hoge-bin = "hoge-lib hoge-conf"

レシピの作成－依存関係について

foo.deb

build

DEPENDS

bar.bbfoo.bb

hoge-lib.deb

build /bin/bar

hoge-conf.deb

RDEPENDS

RDEPENDS

build

hoge-bin.deb

RDEPENDS

DEPENDS = "bar"

PACKAGES = "hoge-bin hoge-lib hoge-conf"

RDEPENDS_hoge-bin = "hoge-lib hoge-conf foo"

hoge.bb

hoge.bb

レシピの作成（例） hello.bb

 ソースコード

 下記3ファイルを含む「hello.tar.gz」を作成

 hello.c

 Makefile

 COPYING: ライセンス表記（GPLv2）

#include <stdio.h>

int main()

{

printf("hello¥n");

return 0;

}

default: clean hello

hello: hello.o

clean:

rm -f hello *.o



28

レシピの作成（例） hello.bb

 hello.bb

 bitbake内部変数

 ${S} (Source Directory): 展開済みソースディレクトリ

 ${D} (Destination Directory): インストール先ディレクトリ

 $ bitbake hello

→ ターゲットバイナリhelloがクロスコンパイルされる

LICENSE = "GPLv2"

LIC_FILES_CHKSUM = ¥

"file://COPYING;md5=7c0d7ef03a7eb04ce795b0f60e68e7e1"

SRC_URI = "http://www.server.com/hello.tar.gz"

do_install() {

install -d ${D}/${bindir}

install -m 0755 ${S}/hello ${D}/${bindir}

}

Pokyをカスタマイズする

 レシピの修正

 ソースコードをバージョンアップ

 カーネルレシピの作成

 自分で用意したカーネルを使う

 BSPの追加

 好きなボードで動くLinuxを作る

レシピの修正

 バージョン変更への要求

 最新バージョンを使いたい

 互換性のため古いバージョンを使いたい

 Pokyのバージョン対応状況

 1レシピが対応するバージョン数は少ない（ほぼ1:1）

 目的のバージョンがない→ レシピを追加

 修正手順

1. レシピのコピー＆バージョン修正

2. バージョン依存部の修正

3. 付属ファイルの修正

レシピの修正 ～ レシピのコピー＆バージョンの修正

 busybox.inc

 バージョンを含まないファイルは変更不要

 busybox_${PV}.bb

 バージョン依存のレシピ記述（要修正）

 busybox-${PV}

 バージョン依存のパッチ、defconfig等（要修正）

meta/recipes-core/busybox/

files/

busybox.inc

busybox_1.18.5.bb

busybox-1.18.5/

busybox_1.20.2.bb

busybox-1.20.2/

バージョン依存のレシピ／付属ファイル

コピー＆バージョン変更

レシピの修正 ～バージョン依存部の修正

 内部変数${PV}はファイル名から自動更新される

 ハードコードされた値・処理は手動で修正
 checksum

 configure／makeオプション

 インストール処理

 注意点
 他のレシピとの依存関係、互換性を維持する必要がある

 ベースにしたバージョンからの差が大きいほど問題

meta/recipes-core/busybox/busybox_1.20.2.bb

require busybox.inc

PR = "r1"

SRC_URI = " ¥

http://www.busybox.net/downloads/busybox-${PV}.tar.bz2;name=tarball ¥

file://udhcpscript.patch ¥

...

file://defconfig"

SRC_URI[tarball.md5sum] = "..."

SRC_URI[tarball.sha256sum] = "..."

自動で更新

手動で更新

レシピの修正 ～付属ファイルの修正

 付属ファイル

 パッチ

 Poky用の修正（パスなど）

 バグfix

 defconfig

 設定ファイル（*.conf）

 パッチは修正コスト大

 busybox_1.20.2の場合：2個のパッチを修正

 数千行規模のパッチが当たらなくなることも...

 複雑さ次第ではメンテナレベルのノウハウが必要（力技）



35

カーネルレシピの作成

 Pokyにおけるカーネルのビルド

 デフォルトではYocto Projectのリポジトリを使用

 自分のカーネルをビルドしたい → カーネルレシピを自作

 BeagleBoard向けの独自カーネルを使う例

 linux-libc-headers-mybeagle.bbを作成

 カーネルヘッダだけをPoky内部にインストールするレシピ

 linux-libc-headers-yocto_git.bbをベースに作成

 linux-mybeagle.bbを作成

 ターゲット向けカーネルイメージをビルドするレシピ

カーネルレシピの作成 ～ linux-libc-headers-mybeagle

recipes-kernel/linux-libc-headers/linux-libc-headers-mybeagle.bb

LICENSE = "GPLv2"

LIC_FILES_CHKSUM = "file://COPYING;md5=d7810fab7487fb0aad327b76f1be7cd7"

INHIBIT_DEFAULT_DEPS = "1"

PROVIDES = "linux-libc-headers"

RPROVIDES_${PN}-dev = "linux-libc-headers-dev"

RPROVIDES_${PN}-dbg = "linux-libc-headers-dbg"

DEPENDS += "unifdef-native"

SRC_URI = "git://localhost/linux.git;protocol=git"

SRCREV = "v3.3.7-mybeagle"

S = "${WORKDIR}/git"

KARCH = "arm"

do_configure() {

oe_runmake allnoconfig ARCH=${KARCH}

}

do_compile () {

:

}

do_install() {

oe_runmake headers_install ¥

INSTALL_HDR_PATH=${D}${exec_prefix} ARCH=${KARCH}

}

BBCLASSEXTEND = "nativesdk"

gitリポジトリを指定

チェックアウト対象のブランチ／タグを指定

カーネルレシピの作成 ～ linux-mybeagle

 kernel.bbclass

 configure, compile, install, package, deployを定義

 ${WORKDIR}/defconfigが.configとして使用される

 動的にmenuconfigを実行＆編集する

 bitbake -c menuconfig virtual/kernel

recipes-kernel/linux/linux-mybeagle.bb

LICENSE = "GPL"

LIC_FILES_CHKSUM = "file://COPYING;md5=d7810fab7487fb0aad327b76f1be7cd7"

inherit kernel

SRC_URI = "git://localhost/linux.git;protocol=git"

SRCREV = "v3.3.7-mybeagle"

S = "${WORKDIR}/git"

do_configure_prepend() {

cp ${S}/config.mybeagle ${WORKDIR}/defconfig

}

linux-libc-headers-mybeagle.bbと同じ

作成済みの.configを使う場合

カーネルビルド用のタスク定義をインポート

BSPの追加

 特定ボード向けのLinux環境一式を作りたい→ BSPを自作

 meta-beagleboard

 meta-cubox

 meta-raspberrypi

 etc.

 BeagleBoard向け独自BSPを作る場合

1. meta-mybeagleを作成

 meta-skeletonをベースに

2. BSP設定ファイルの作成

3. マシン設定の作成

4. レシピのオーバライド

5. レイヤ設定の修正

BSPの追加 ～ BSP設定ファイルの作成

 BSP名

 レシピの優先度（BBFILE_PRIORITY_*）

 値の高いBSPに属するレシピが優先される

 ターゲットマシンの決定（MACHINE）

 conf/machine/${MACHINE}.confが使用される

meta-mybeagle/conf/layer.conf

BBPATH := "${BBPATH}:${LAYERDIR}"

BBFILES := " ${BBFILES} ¥

${LAYERDIR}/recipes-*/*/*.bb ¥

${LAYERDIR}/recipes-*/*/*.bbappend ¥

"

BBFILE_COLLECTIONS += "mybeagle"

BBFILE_PATTERN_mybeagle := "^${LAYERDIR}/"

BBFILE_PRIORITY_mybeagle = "6"

MACHINE = "mybeagle" ターゲットマシン

BSP名

レシピ優先度

BSPの追加 ～マシン設定の追加

 CPU設定
 march等のCPU依存の設定（conf/machine以下の既存ファイルを流用）

 カーネル設定
 カーネルレシピ（virtual/kernel）にmybeagle用カーネルを指定

 出力イメージの指定（zImage, uImage）

 rootfs設定
 出力イメージの指定（tar ball, ext3, jffs2, cpio, etc.）

 シリアルコンソール設定（inittabに反映）

 U-Boot設定
 使用するconfig、ロードアドレスの指定（u-bootレシピの内部変数）

meta-mybeagle/conf/machine/mybeagle.conf

require conf/machine/include/tune-cortexa8.inc

PREFERRED_PROVIDER_virtual/kernel = "linux-mybeagle"

PREFERRED_PROVIDER_linux-libc-headers = "linux-libc-headers-mybeagle"

KERNEL_IMAGETYPE = "uImage"

IMAGE_FSTYPES += "tar.bz2 ext3"

SERIAL_CONSOLE = "115200 ttyO2"

EXTRA_IMAGEDEPENDS += "u-boot"

UBOOT_MACHINE = "omap3_beagle_config"

UBOOT_ENTRYPOINT = "0x80008000"

UBOOT_LOADADDRESS = "0x80008000"

CPU設定

カーネル設定

rootfs設定

U-Boot設定

BSPの追加 ～ レシピのオーバライド

meta-mybeagle/recipes-connectivity/openssh/openssh_5.8p2.bbappend

do_install_append() {

sed -i "s|.*¥(PermitRootLogin¥) yes|¥1 no|g" ¥

${D}/${sysconfdir}/ssh/sshd_config

}

meta-mybeagle/recipes-core/images/core-image-minimal.bbappend

IMAGE_INSTALL += "task-core-ssh-openssh"

IMAGE_DEVICE_TABLES += "files/device_table-mybeagle.txt"

meta-mybeagle/files/device_table-mybeagle.txt

sshd_configの修正

・opensshをrootfsに追加
・静的デバイスファイルの一覧表を指定

# path       type      mode   uid     gid   major   minor   start     inc   count

/dev            d       755     0       0       - - - - -

/dev/console    c       662     0       0       5       1       - - -

/dev/kmem       c       640     0       15      1       2       - - -

/dev/mem        c       640     0       15      1       1       - - -

/dev/null       c       666     0       0       1       3       - - -

...

/dev/tty        c       600     0       5       4       0       0       1       7

/dev/ttyO2      c       640     0       5       253     2       - - -

/dev/mmcblk0    b       660     0       6       179     0       - - -

/dev/mmcblk0p   b       660     0       6       179     1       1       1       5

ボード依存のデバイスファイルを追加

BSPの追加 ～ レイヤ設定の修正＆ビルド

 bblayers.conf.sample

 デフォルトのレイヤ設定

 BBLAYERS: 有効にするBSP一覧

 ビルド

 bitbake core-image-minimal

 layer.conf、マシン設定、レシピのオーバライドが適用された状態で
ターゲット向けイメージがビルドされる

meta-yocto/conf/bblayers.conf.sample

LCONF_VERSION = "4"

BBFILES ?= ""

BBLAYERS = " ¥

##COREBASE##/meta ¥

##COREBASE##/meta-yocto ¥

##COREBASE##/meta-mybeagle ¥

"

・BSPを追加（複数追加可）
・meta, meta-yoctoはデフォルト

Pokyを利用する上での問題点

ビルド時間

 ビルド時間が長い

 早くても30分程度

 マシン性能が悪いと半日かかることも

 主な原因

 native用ツール、クロスツールチェインを含むフルビルド

 依存関係の冗長性

 configure次第では不要となる依存関係もある

 ビルド依存先のレシピから連鎖的に拡大

ビルド時間の具体例

 ビルド時間の計測

ビルド対象 レシピ数 概要

rootfs-A 84 busyboxベースの最小rootfs＋カーネル

rootfs-B 145 (A) + udev、sshd、Qtなど

ビルド環境: Intel Xeon E3-1270(3.4GHz x4) / 24GB RAM / 2TB 7200rpm HDD

スペック以上は
頭打ち

パッケージ数が多いほど
並列化効率は上がる

ディスク消費量

 ビルドディレクトリがディスクを圧迫

 主な原因
 native用ツール、クロスツールチェインを含むフルビルド（前述）

 依存関係の冗長性（前述）

 ビルド過程のファイルがすべて残っている

 ダウンロードファイル、展開後のソース、バイナリ、パッケージファイル

 対策
 他のレシピから参照されない構築後のファイルを自動削除

 ターゲット設定に完全に依存しない部分を共有したビルド方法

 downloadデータ、native用ツールのワークディレクトリなど

 レシピ依存関係の改善

ビルド対象 レシピ数 ビルド後の容量

rootfs-A 84 11GB

rootfs-B 145 19GB

busyboxのみ 55 8.9GB

まとめ

 Pokyの概要説明

 内部構造（ビルドシーケンス、ディレクトリ構造）

 サンプルレシピの記述方法

 Pokyを使用する主なメリット

 クロスビルド環境を自動的に構築

 BSPの活用による開発効率向上

 Pokyのカスタマイズ方法

 バージョン変更

 カーネルレシピの作成

 BSPの追加

 課題

 ビルド時間

参考URL

 Yocto Project

 http://www.yoctoproject.org/

 Poky Reference Manual

 http://www.yoctoproject.org/docs/1.1/poky-ref-manual/poky-ref-

manual.html

 Please read docs under version "1.1" (base of poky-skerlet)

 OpenEmbedded User Manual

 http://docs.openembedded.org/usermanual/usermanual.html

 BitBake User Manual

 http://docs.openembedded.org/bitbake/html/

